Trinity College – Hartford
Engineering Department
ENGR-323-01 – Embedded Systems Design

8051 Embedded System Heart Beat (BPM) Display

Liu Restrepo Sanabria
Fadhil Ahmed

liu.restreposanabria@trincoll.edu,
fadhil.ahmed@trincoll.edu

Taikang Ning, Ph.D.

April, 2025

Contents

1 Problem Statement

Design and implement a digital heart rate monitor using the 8051 microcontroller. The system reads the heartbeat of the user via digital pulses generated from an analog signal that has been conditioned using an operational amplifier (op-amp) circuit. This op-amp circuit cleans the raw heartbeat waveform and produces sharp digital pulses suitable for triggering the external interrupt of the microcontroller.

The microcontroller calculates the heart rate in beats per minute (BPM) based on the time interval between consecutive pulses. These BPM values are then displayed on a 4-digit 7-segment display using a time-division multiplexing (TDM) approach. The program, written in Assembly language, handles timing, division, scaling, and digit decoding to produce real-time BPM values for display.

This project emphasises low-level embedded programming using the 8051 microcontroller, real-time multiple interrupt handling, and analog-to-digital signal interfacing using discrete electronics.

2 Specific Design Goal

The objective of this laboratory project is to design an embedded system based on the 8051 microcontroller that can accurately detect, calculate, and display the heart rate of a user in beats per minute (BPM.) To achieve this, the system must fulfill the following specific design goals:

- Use an operational amplifier circuit to condition an analog heartbeat signal into clean digital pulses.
- Utilise the external interrupt 0 (`INT0`) of the 8051 to detect falling edges corresponding to heartbeats.
- Employ Timer 1 to measure the time between two consecutive pulses, enabling the calculation of BPM.
- Scale and process the timing data using 8-bit arithmetic operations to compute BPM values efficiently.
- Use Timer 0 to implement a time-division multiplexing (TDM) scheme for refreshing a 4-digit 7-segment display.
- Convert the BPM value into individual digits and map them to their corresponding display codes for real-time visualisation.
- Handle all logic and computation using low-level Assembly programming for the 8051 architecture.

3 Theory and Background

3.1 8051 Microcontroller Overview

The 8051 microcontroller is an 8-bit microcontroller developed by Intel, widely used in embedded system applications due to its simple architecture and powerful interrupt and timer features. It includes internal RAM, ROM, I/O ports, timers, and supports multiple interrupt sources. In this project, the 8051 is used to process external pulses derived from the heartbeat signal of a user, calculate beats per minute (BPM,) and display the result on a 4-digit 7-segment display.

3.2 Signal Conditioning with Op-Amp Circuit

The raw heartbeat signal, typically obtained from a sensor such as a photodiode or an analog pulse sensor, contains noise and does not produce clean logic-level transitions. To make it suitable for digital processing, the analog signal is passed through an op-amp comparator circuit. This circuit sharpens the waveform and produces well-defined digital pulses on each heartbeat. These pulses are then connected to the external interrupt pin of the 8051 (specifically `P3.2` / `INT0`.)

3.3 External Interrupt Handling

The 8051 microcontroller architecture provides two external interrupt sources: `INT0` (pin P3.2) and `INT1` (pin P3.3.) These interrupts enable the microcontroller to respond immediately to external asynchronous events without the need for continuous polling [1].

Each external interrupt can be configured in one of two modes: level-triggered or edge-triggered. In level-triggered mode, the interrupt remains active as long as the corresponding pin is held low. In edge-triggered mode, the interrupt is activated by a negative edge (high-to-low transition) on the interrupt pin. The trigger configuration is controlled by the Interrupt Type Control (ITx) bits within the `TCON` register, where setting `IT0` or `IT1` to logic high configures `INT0` or `INT1` respectively to operate in edge-triggered mode [1].

External interrupts are commonly used in embedded systems for event-driven processing such as pulse detection, signal edge capture, or interfacing with peripheral hardware.

3.4 Time-Division Multiplexing (TDM)

Due to limited I/O pins on the 8051, all four 7-segment displays cannot be driven simultaneously. Instead, the displays are refreshed one at a time at a high frequency using Timer 0. This process is known as Time-Division Multiplexing (TDM.) A demultiplexer

(e.g., 74LS138) selects one display at a time, and the decoded digit is sent to the BCD-to-7-segment decoder (e.g., 7447.) When performed fast enough, this sequential refresh creates the illusion that all digits are continuously illuminated.

3.5 Design Constraints Calculations

Microcontroller-based systems that employ multiple interrupts must be carefully designed to ensure correct and efficient operation under real-time constraints. The primary challenges in designing such systems involve balancing CPU utilisation, interrupt latency, and system responsiveness [2].

3.5.1 CPU Utilisation

In interrupt-driven systems, CPU utilisation is directly influenced by the frequency and execution time of each Interrupt Service Routine (ISR.) The total CPU load imposed by all interrupts can be approximated by

$$U_{\text{CPU}} = \sum_{k=1}^{N} \frac{T_k}{\text{TP}_k} < 100\% \tag{1}$$

where TP_k is the time period of interrupt k and T_k is the execution time of the corresponding ISR.

3.5.2 Latency Requirement

Interrupt latency is the time elapsed between the occurrence of an interrupt request and the execution of its corresponding ISR. In systems with nested or multiple interrupts, latency can increase due to higher-priority ISRs preempting lower-priority interrupts, or due to global interrupt disabling within critical sections [2].

The latency requirement is calculated as follows

$$L_i = T_{i+} + \sum_{k=1}^{N-1} N_k T_k < (\text{TP}_i - T_i) \tag{2}$$

where N_k can be calculated as shown later in (3) and it represents the number of times the ISR of higher priorities (k interrupts of a higher priority) can be served before i is served. T_{i+} is the largest time taken by an interrupt with a lower priority than i.

$$N_k = \text{INT}\left(\frac{\text{TP}_i - T_i}{\text{TP}_k}\right) + 1 \tag{3}$$

4 Design Strategy

4.1 Hardware Overview

The system consists of several integrated components that work together to detect, process, and display the heartbeat of the user:

- **8051 Microcontroller**: Acts as the central processing unit that handles pulse detection, BPM computation, and display control.
- **Op-Amp Comparator Circuit**: Conditions the raw analog heartbeat signal to generate clean digital pulses. This circuit removes noise and produces a rising edge at logic level for each heart pulse.
- **BCD to 7-Segment Decoder (7447)**: Converts Binary-Coded Decimal (BCD) digits into segment patterns suitable for the 7-segment displays.
- **4-Digit 7-Segment Display (LTC-5653G-01)**: Displays the computed BPM value. The digits are refreshed sequentially via time-division multiplexing.
- **Demultiplexer (74LS138)**: Selects one of the four 7-segment digits at a time, allowing shared use of data lines.
- **EEPROM (2816)**: Stores the 8051 assembly code, which is loaded at boot to execute the BPM measurement and display logic.

The 8051 hardware architecture utilised in previous designs is recycled, but the connection to `P3.2` changes. A comparator circuit consisting of an operational amplifier (LM321,) a potentiometre and a signal input (containing the EKG) will provide an output of digitised pulses that will be used as external interrupts to tell the 8051 a pulse has been detected. See Fig. 1 for the comparator diagram.



Figure 1: Operational Amplifier comparator circuit for pulse output schematic

The digital output from the op-amp circuit is connected to the external interrupt pin `P3.2` (`INT0`) of the 8051. The data lines from the microcontroller are shared between all digits and driven through the 7447 decoder, while digit selection is controlled using the demultiplexer. A standard 12 MHz crystal oscillator provides the clock source. Fig. 2 shows the internal connections between the 8051. Note that this is a recycled schematic from laboratory one, so the connections in pin `P3.2` are different in this design.



Figure 2: 8051 hardware architecture with multiplexed display and external EEPROM

The components used for the design are listed in Table 1.

4.2 Resolution and Accuracy

Considering that this project employs multiple interrupts and the limitation of 8-bit division (not 16-bit division,) there are resolution and accuracy constraints that have to adhere to the limitations for an efficient software design.

Given that only an 8-bit division can be performed, the result of the division is limited to 255. However, that is an acceptable limit given that 100 BPM at rest is tachycardia, so the values will never approach 255 BPM in a realistic situation. For calculating the heartbeat in BPM, the time between two pulses will be taken and averaged over one minute. In other words, the following equation will be employed to obtain the heartbeat:

$$\mathrm{BPM} = \frac{60\,000\,\mathrm{ms}}{N_{\mathrm{TF1}} \times \mathrm{TF1}}, \tag{4}$$

Table 1: List of Components Used in the Heart Rate Display

Component	**Quantity**
330Ω resistors	7
4-digit 7-segment display (LTC-5653G-01)	1
SN74LS04 (NOT gate)	1
8051 microcontroller	1
7447 (BCD-to-7-segment decoder)	1
2k x 8 EEPROM (AT28C16)	1
HC573 (Latch)	1
74138 (Demultiplexer)	1
Push buttons	1
1kΩ resistors	1
22pF ceramic capacitors	2
47μF electrolytic capacitor	1
CADET Board	1
Potentiometre (included in CADET Board)	1
LM 741 (op-amp)	1

where 60 000 ms is one minute in milliseconds, N_{TF1}, is the number of timer 1 overflow occurrences, and TF1 is the period in which timer 1 overflow occurs in milliseconds.

Nevertheless, (4) fails to do 8-bit division since the dividend is 60 000 and it should be limited to 255. Hence, the value should be scaled down to 250, which is close enough, but not exactly, 255 for uncertainty considerations. This scaling factor will play a big role in the accuracy of the BPM because if it is a big number then it will affect the value displayed given that multiplication (to scale back up) can only be 8-bit as well as division.

Hence, TF1 has to be selected so that the scaling factor is a reasonable number (such as 5) and the worst-case scenario (where the numerator is 250) is considered to define what the value of TF1 will be. Hence,

$$\begin{aligned} \text{TF1} &= \frac{60\,000\,\text{ms}}{250 \times 5} \\ &= \frac{60\,000\,\text{ms}}{1\,250} \\ &= 48\,\text{ms} \end{aligned}$$

Note that if the scaling factor had been $k = 4$ (better resolution,) the value of TF1 would have been 60 ms. This value is close to 65 ms, which is almost the maximum value of timer overflow given a 12 kHz crystal oscillator (65 ms = 65 000 μs $\sim$ 65 536 μs,) and a high value of timer overflow period would have introduced a higher error when it comes to receiving the next pulse (so the accuracy would have decreased.) Hence, the scaling factor is chosen to be $k = 5$ and the timer overflow period to be TF1 = 48 ms. Hence,

the formula used to calculate the heartbeat in BPM is

$$\frac{\text{BPM}}{k} = \frac{60\,000}{N_{\text{TF1}} \times \text{TF1}}$$

where $k = 5$, and $\text{TF1} = 48\,\text{ms}$. Thus,

$$\frac{\text{BPM}}{5} = \frac{250}{N_{\text{TF1}}} \tag{5}$$

N_{TF1} is limited to 255 as well since it is an 8-bit counter in one of the registers, but it is guaranteed to be under 255 because if $N = 250$, then $\text{BPM} = 5$, which is the minimum heartbeat detected by this design.

4.3 Software Design

This program calculates and displays the heart rate in Beats Per Minute (BPM) based on digitalised EKG pulses. The design uses Timer 0 for a multiplexed 7-segment display, Timer 1 for measuring the time between pulses, and an external interrupt (`INT0`) to detect each EKG pulse. The full working code can be found in Appendix A.

4.3.1 Interrupt Vector Table and Entry Point

Listing 1: Interrupt Vector Table and Main Entry

```
1  ORG 0000h
2  AJMP MAIN
3
4  ORG 03h
5  AJMP ISREX0     ; External interrupt 0 handler
6
7  ORG 0Bh
8  AJMP ISRTF0     ; Timer 0 overflow handler
9
10 ORG 1Bh
11 AJMP ISRTF1     ; Timer 1 overflow handler
```

This block sets the interrupt vectors as follows:

- `INT0 (EX0)`: Used for pulse detection. The pulse comes from the output of the op-amp signal conditioning circuit.
- `TF0`: Handles the TDM update for the 7-segment display.
- `TF1`: Measures the time elapsed between two pulses.

4.3.2 Main Initialisation Routine

Listing 2: Initialisation of Timers, Variables, and Interrupts

```
1  MAIN:
2      MOV TMOD, #11h          ; Timer 0 & 1 in Mode 1 (16-bit)
3      MOV TH0, #0ECh
4      MOV TL0, #07Ch          ; Timer 0 for 5ms delay
5
6      MOV TH1, #044h
7      MOV TL1, #07Fh          ; Timer 1 for 48ms delay
8
9      SETB P3.2               ; sets INT0 to be an input
10
11     SETB ET0
12     SETB TR0                ; Enable Timer 0
13
14     SETB EX0
15     SETB IT0                ; Edge-triggered INT0
16
17     SETB ET1                ; Enable Timer 1 overflow (not started)
18     SETB EA                 ; Enable global interrupts
```

This block initialises both timers for the purpose mentioned before. Timer 1 is not running (i.e. there is no `SETB TR1`) because it should start running if and only if a pulse is detected. Additionally, the external interrupt is configured as edge-triggering (falling edge) for EKG pulse detection.

4.3.3 Variable Initialisation

Here, BPM is broken down into digits and given an address and an initial value for each digit. `THOUSANDS_BPM` will never change, but it serves as a placeholder to maintain consistency over the design. `R0` iterates through the digit memory addresses for display, whereas `R6` counts Timer 1 overflows between two detected pulses.

Listing 3: Initialisation of Display Digits and Counters

```
1      ONES_BPM     EQU 30h
2      TENS_BPM     EQU 31h
3      HUNDREDS_BPM EQU 32h
4      THOUSANDS_BPM EQU 33h   ; Always #30h
5
6      BPM EQU 40h
7
8      MOV ONES_BPM, #00h
9      MOV TENS_BPM, #10h
10     MOV HUNDREDS_BPM, #20h
11     MOV THOUSANDS_BPM, #30h
```

```
12
13      MOV BPM, #00
14      MOV R0, #30h            ; R0 as display pointer
15      MOV R6, #00             ; Timer 1 overflow counter
16
17  WAIT:
18      NOP
19      SJMP WAIT
```

4.3.4 Timer 0 ISR: Display Refreshing

This routine refreshes one digit of the 7-segment display per call, implementing time-division multiplexing (TDM.) It cycles through memory locations `30h` to `33h`. The critical region here is defined and it lasts three instruction cycles (i.e. $\sim 3\,\mu$s) with a period of 5 ms, which is the same as the period for `TF0`.

Listing 4: Timer 0 ISR – Display Multiplexing

```
1  ISRTF0:
2      CLR EA
3      CLR TF0
4      MOV TH0, #0ECh
5      MOV TL0, #07Ch
6      SETB EA
7
8      MOV P1, @R0
9      INC R0
10     CJNE R0, #34h, DONE
11     MOV R0, #30h
12
13 DONE:
14     NOP
15     RETI
```

4.3.5 External Interrupt 0 ISR: Pulse Detection

When a pulse is detected, if Timer 1 is already running, compute BPM. Otherwise, start timing the interval. This helps handle the first interruption which occurs to activate Timer 1.

Listing 5: External Interrupt ISR – Pulse Triggered

```
1  ISREX0:
2      JB TR1, CALCULATE_TIME  ; If Timer 1 running, calc BPM
3      SETB TR1                ; Otherwise start it
4      SJMP RESET_COUNTER
```

4.3.6 `CALCULATE_TIME` Subroutine

This subroutine implements the logic developed in Section 4.2, successfully implementing 8-bit division and multiplication, considering the remainder to scale back up the resulting BPM accurately. First, Timer 1 has to be stopped, and save the value of `R6`, which is N_{TF1}, into the accumulator to proceed to `GET_BPM` which will divide 250 by `R6` and store the quotient into `R3` and the remainder into `R4`. Hence,

$$\begin{aligned} N_{\text{TF1}} &= \text{R6} \\ Q &= \left\lfloor \frac{250}{N_{\text{TF1}}} \right\rfloor \\ R &= 250 - Q \times N_{\text{TF1}} \\ \Rightarrow Q &\rightarrow \texttt{R3} \\ \Rightarrow R &\rightarrow \texttt{R4} \end{aligned}$$

Now, both the quotient and the remainder must be scaled and added to successfully obtain the desired BPM. First, scale back the quotient and save it in `R1`, then scale back the remainder and divide it by `R6` to finally add it to the scaled up quotient.

$$\begin{aligned} Q_{\text{scaled}} &= Q \times 5 \\ R_{\text{scaled}} &= \left\lfloor \frac{R \times 5}{N_{\text{TF1}}} \right\rfloor \\ \text{BPM} &= Q_{\text{scaled}} + R_{\text{scaled}} \end{aligned} \tag{6}$$

The resulting sum will be saved in the variable `BPM` and `DECODE` is called which will separate the BPM into digits. Finally, jump to `RESET_COUNTER` to reset the counter and start the next calculation.

Listing 6: BPM Calculation Routine

```
1  CALCULATE_TIME:
2      CLR TR1
3      MOV A, R6
4  GET_BPM:
5      MOV B, A
6      MOV A, #250
7      DIV AB
8      MOV R3, A
9      MOV R4, B
10
11     MOV B, #05
12     MUL AB
13     MOV R1, A
14
```

```
15      MOV A, R4
16      MOV B, #05
17      MUL AB
18
19      MOV B, R6
20      DIV AB
21      ADD A, R1
22      MOV BPM, A
23
24      ACALL DECODE
25      SJMP RESET_COUNTER
```

4.3.7 Resetting the Counter

After each BPM update, the overflow counter is reset to measure the next interval.

Listing 7: Reset Pulse Counter

```
1 RESET_COUNTER:
2      MOV R6, #00
3      SETB TR1
4      RETI
```

4.3.8 BPM Digit Decoding

To decode the value of BPM to get each digit to display individually the ones, tens, and hundreds of BPM, the value of BPM is divided by 100, the quotient is the number of hundreds of BPM in the heartbeat, whereas the remainder will be used to calculate the tens of BPM. The same process is done to obtain the tens of BPM and ones of BPM.

In the process, add offsets so that each digit corresponds to its correct 7-segment representation. In other words, since `HUNDREDS_BPM` is digit 2, then add `0x20` to shift the number to the corresponding digit. For `TENS_BPM` it would be adding `0x10`.

Listing 8: Decoding BPM into Digits

```
1 DECODE:
2      MOV A, BPM
3      MOV B, #100
4      DIV AB
5      ADD A, #20h
6      MOV HUNDREDS_BPM, A
7
8      MOV A, B
9      MOV B, #10
10     DIV AB
11     ADD A, #10h
12     MOV TENS_BPM, A
```

```
13
14      MOV A, B
15      MOV B, #1
16      DIV AB
17      MOV ONES_BPM, A
18      RET
```

4.3.9 Timer 1 ISR: Interval Measurement

Each overflow of Timer 1 (48 ms) increments `R6` and reloads the timer with its corresponding values. This counter is used in the BPM calculation when a new pulse arrives.

Listing 9: Timer 1 ISR – Counting 48ms Overflows

```
1  ISRTF1:
2       CLR TR1
3       CLR TF1
4       CLR EA
5
6       INC R6                  ; Increment overflow counter
7
8       MOV TH1, #044h
9       MOV TL1, #07Fh
10
11      SETB EA
12      SETB TR1
13      RETI
```

4.4 Design Constraints

4.4.1 CPU Utilisation

The system used three independent interrupts. By counting the worst-case scenario of instruction cycles for each interrupt, a calculation can be done employing (1), thus the period of each timer is already defined, whereas the period of the external interrupt (heartbeat) varies depending on the user. In the worst case, the user will have a heart rate of 180 BPM, which is 3 BPS. Hence, $T_{\text{EX0}} \sim 334$ ms. Table 3 shows the execution time for each ISR and its period. Critical Region 1 is used to reload Timer 0 and avoid re-entrant access during display refresh. Critical Region 2 is used to increment `R6` (pulse interval counter) without interruption and reload Timer 1.

By using the values shown in Table 3, the CPU utilisation is computed as follows:

$$\begin{aligned} U_{\text{CPU}} &= \frac{87\,\mu\text{s}}{334\,\text{ms}} + \frac{19\,\mu\text{s}}{5\,\text{ms}} + \frac{12\,\mu\text{s}}{48\,\text{ms}} \\ &= 0.00431 \\ &= 4.31\% \end{aligned}$$

4.4.2 Latency Requirement

For the latency requirement, each interrupt has to be evaluated to see whether this requirement is met. The required parameters to evaluate it are shown in Table 2 and the calculation is computed for L_i and $\text{TP}_i - T_i$ to check if the latency requirement is met.

Parameter	External Interrupt ($i = 1$)	Timer 0 overflow ($i = 2$)	Timer 1 overflow ($i = 3$)
T_{i+}	19 μs	12 μs	7 μs
TP_i	334 ms	5 ms	48 ms
T_i	87 μs	19 μs	12 μs
N_1	-	1	1
T_1	-	87 μs	87 μs
N_2	-	-	1
T_2	-	-	19 μs
L_i	19 μs	99 μs	113 μs
$\text{TP}_i - T_i$	333.981 ms	4.981 ms	47.988 ms

Table 2: Latency Requirement calculation table

Table 2 shows the parameters needed to calculate the latency requirement and, as per (2), it is met given that $L_i < \text{TP}_i$ for $i = 1, 2, 3$ which correspond to each interrupt service routine.

Interrupt	Time Required [μs]	Time Period [ms]
External Interrupt 0	87	334
Timer 0 Overflow	19	5
Timer 1 Overflow	12	48
Critical Region 1	7	**N/A**
Critical Region 2	7	**N/A**

Table 3: Period and time required to execute the ISR of each interrupt employed in the system

5 System Implementation

5.1 Wiring Process

The hardware implementation for this laboratory reused the same configuration established in Laboratory 1 and 2, with the addition of an op-amp pulse conditioning circuit. The entire system was assembled on a breadboard, following consistent pin mappings and layout for seamless integration with the 8051 development board.

The following components were used:

- **8051 Development Board**: Functioned as the core processor. Port 1 was used exclusively for both digit data output and digit selection signals. Port 3 was used only for the external interrupt input `P3.2` (`INT0`.)
- **Op-Amp Comparator Circuit**: Newly introduced in this laboratory, this circuit converted the analog heartbeat waveform into sharp digital pulses. Powered with a dual ±5V supply, its output was connected directly to `P3.2`, triggering interrupts on rising edges.
- **BCD to 7-Segment Decoder (7447)**: Connected to the lower bits of Port 1, this decoder translated BCD values into segment control signals for the 7-segment display.
- **Demultiplexer (74LS138)**: Also connected to Port 1, the demultiplexer selected which digit of the display was active during time-division multiplexing. Digit selection and segment data shared the same port using a time-sliced logic scheme.
- **4-Digit 7-Segment Display (Common Cathode)**: Each digit was enabled one at a time through the demultiplexer, while the decoder drove the segment lines. This minimised the number of active outputs and reduced pin usage.
- **EEPROM (2816)**: Stored the program written in 8051 assembly and provided non-volatile code storage for the microcontroller.
- **Clock Crystal (12 MHz)**: Supplied the timing reference for the microcontroller, enabling accurate timer operation for both BPM measurement and display updates.

Power rails were stabilised using decoupling capacitors across major ICs, and pull-down resistors were used on the interrupt input to suppress spurious triggering. The consistent use of Port 1 for both data and control signals reflected a compact and efficient design choice for limited I/O environments.



Figure 3: Op-Amp Comparator Circuit for EKG Signal Conditioning



Figure 4: Complete functioning circuit with 8051 system and EKG detector

5.2 System Validation

The system was validated through a combination of functional testing and signal analysis using laboratory equipment:

- A signal generator was used to produce a known, periodic pulse pattern simulating a human heartbeat. This ensured consistency and repeatability during testing.
- An oscilloscope was used to examine both the raw signal and the cleaned, digital output from the op-amp comparator circuit. This confirmed that the comparator reliably converted noisy analog input into sharp digital transitions suitable for interrupt triggering.

- The cleaned signal from the op-amp was verified to rise sharply to 5V upon each simulated pulse, with consistent timing and no intermediate noise or bouncing.
- The output BPM was cross-validated using manual stopwatch measurements to confirm the accuracy of the system. Across several test cases, the displayed BPM matched expected values within ±1 BPM for rates between 50–120 BPM.
- The 7-segment display updated smoothly via time-division multiplexing (TDM,) showing no visible flicker or digit artifacts.
- Long-term operation (multi-minute runtime) demonstrated system stability, with no drift in BPM calculation or missed pulse events.

The combination of synthetic signal input, oscilloscope verification, and accurate visual output confirmed that the system functioned reliably and met all design expectations. The use of an interrupt-driven architecture contributed significantly to the responsiveness of the system and precision during operation.

6 Results and Discussion

6.1 Performance Analysis

The heart rate monitoring system demonstrated excellent performance during testing. Upon receiving a clean digital pulse input from the op-amp circuit, the 8051 microcontroller responded with high reliability and minimal latency. External interrupts were consistently triggered without false positives or missed detections, showcasing the robustness of the edge-triggered configuration on `INT0`.

The calculated BPM values matched manual timing benchmarks with high accuracy. For instance, when input pulses were generated at a consistent 1-second interval (60 BPM,) the system output stabilised around 60 BPM within one to two cycles. At faster rates (e.g., 85 BPM,) the system maintained correct readings, with fluctuations within ±1 BPM — which is well within acceptable biological and digital tolerance for such low-resolution timing.

The time-division multiplexing (TDM) display system, driven by Timer 0 interrupts, was visually stable and free of perceptible flicker. Each digit on the 7-segment display updated fluidly, and the decoder-multiplexer combination ensured that values were rendered clearly in real-time. Even during rapid heartbeat transitions, the display updated almost instantly, demonstrating a tight feedback loop between pulse detection, BPM calculation, and visual output.

Notably, the performance of the system was unaffected by background processing since it relied almost entirely on interrupts. This ensured that BPM calculation remained consistent regardless of timing irregularities in other parts of the program.

6.2 Efficiency Considerations

This project was designed with hardware and computational efficiency in mind, leveraging several key strategies:

- **Interrupt-Driven Architecture**: By offloading logic to ISRs, the system avoids polling loops, reducing CPU idle time and allowing the main program to remain simple and energy-efficient. This is shown with a CPU utilisation of 4.31%.
- **Optimised Use of Timers**: The clever use of Timer 1 to count overflow intervals, rather than exact ticks, enabled the measurement of BPM without the complexity of 16-bit math, ideal for the 8-bit architecture of 8051.
- **Minimal Instruction Footprint**: All operations were performed using low-level assembly with tight register-level control. This resulted in minimal memory usage and fast execution cycles, even during multiplication and division steps.
- **Segment Display Optimisation**: By preloading digit values into memory and sequentially displaying them via TDM, the system minimised I/O pin usage while maintaining a stable, real-time display. This choice also conserved power by limiting active output lines at any given time.
- **Reduced Latency**: The latency of the system between heartbeat detection and BPM display update was kept extremely low. Table 2 shows that for each ISR, the latency was very low compared to the difference between the period and execution time of the ISR. With ISR routines executing quickly and timers preloaded upon entry, the BPM calculation is completed in a few instruction cycles, and the entire system is optimised.

Overall, the design reflects the strengths of embedded systems engineering: combining precision timing, hardware-aware programming, and resource-conscious design. It achieves accurate BPM monitoring and display without reliance on floating-point arithmetic, lookup tables, or external computation — a testament to the power of thoughtful low-level design.

7 Conclusion

This project successfully demonstrated the design and implementation of a real-time heart rate monitor using the 8051 microcontroller. By combining analogue signal conditioning

through an op-amp comparator with precise digital timing and interrupt handling, the system was able to detect heartbeats and compute BPM values accurately and efficiently.

The use of Timer 1 for interval measurement and Timer 0 for display multiplexing showcased the effective utilisation of the internal hardware resources of the 8051. The decision to use interrupt-driven programming significantly enhanced responsiveness and system stability, allowing real-time BPM calculation without burdening the main execution loop.

Furthermore, the implementation of 8-bit arithmetic to handle scaled BPM calculations highlighted a creative solution to the limitations of the 8051 architecture, avoiding complex 16-bit math while maintaining accuracy.

The final output — a smooth, flicker-free BPM display — validated the integration of hardware and software components. The system remained consistent across a wide range of simulated heart rates, proving to be both robust and scalable.

Overall, this project exemplifies the principles of embedded systems design: low-level control, efficient resource management, and direct hardware interfacing. It not only met the functional objectives but also provided valuable hands-on experience in working with interrupts, timers, and real-time signal processing in a microcontroller-based environment.

References

[1] K. J. Ayala, *The 8051 Microcontroller*. Cengage Learning, 3rd ed., 2010.

[2] P. A. Laplante, *Real-Time Systems Design and Analysis*. Wiley, 4th ed., 2017.

A Heart Beat Display code

Listing 10: 8051 Heartbeat Calculation and Display Assembly Code

```
1           ORG 0000h
2           AJMP MAIN
3
4           ORG 03h
5           AJMP ISREX0
6
7           ORG 0Bh
8           AJMP ISRTF0
9
10          ORG 1Bh
11          AJMP ISRTF1
12
13          ORG 0100h
14  MAIN:
```

```
15          ; Set timer mode with mode 1 for timer 0 and 1
16          MOV TMOD, #11h
17
18          ; Initialise timer 0
19          MOV TH0, #0ECh
20          MOV TL0, #07Ch      ; 5 ms counter for 12 MHz crystal
21
22
23          ; Initialise timer 1
24          MOV TH1, #044h      ; 48 ms delay timer
25          MOV TL1, #07Fh      ; (FFFF - (48,000)d)
26
27          SETB P3.2           ; Put high impendance in P3.2
28
29          ; Enable timer overflow 0
30          SETB ET0        ; Enables timer overflow 0
31          SETB TR0        ; Start timer running
32
33          ; Enable external interrupt
34          SETB EX0        ; Enable external interrupt 0
35          SETB IT0        ; Sets INT0 to be edge-triggering (not level
                -triggered)
36
37          ; Just enable timer overflow 1, not starting timer yet
38          SETB ET1
39
40          ; Enable global interrupts
41          SETB EA
42
43          ; Declaration of variables for display
44          ONES_BPM EQU 30h
45          TENS_BPM EQU 31h
46          HUNDREDS_BPM EQU 32h
47          THOUSANDS_BPM EQU 33h       ; This one is just for reference
                (it will always be #30h)
48
49          ; Declaration of variables for pulses
50          BPM EQU 40h
51
52          ; Initialisation of variables, set everything to zero
53          MOV ONES_BPM, #00h
54          MOV TENS_BPM, #10h
55          MOV HUNDREDS_BPM, #20h
56          MOV THOUSANDS_BPM, #30h
57
58          ; Set number of BPM to 0 by default
59          MOV BPM, #00
```

```
60
61          ; Initialise Registers
62          MOV R0, #30h                ; Move the first digit to do TDM
63          MOV R6, #00                 ; This register will be the
                counter for time interval
64
65          SJMP WAIT                   ; Wait for interruptions
66
67  WAIT:
68          NOP
69          SJMP WAIT
70
71  ;------------------START THE ACTUAL CODE HERE
        ---------------------------------------
72
73  ;------------------TIMER OVERFLOW 0 SERVICE ROUTINE
        --------------------------------
74  ; TDM display for 4-digit 7-segment display
75  ISRTF0:
76          ; Disable global interrupts
77          CLR EA
78
79          ; Clear timer flag and reload timer
80          CLR TF0
81          MOV TH0, #0ECh
82          MOV TL0, #07Ch
83
84          ; Re-enable global interrupts
85          SETB EA
86
87          ; Move into port 1 the value pointed by R0 and increase R0
88          MOV P1, @R0
89          INC R0
90
91          ; Check if R0 is out of bounds
92          CJNE R0, #34h, DONE
93
94          ; If out of bounds, reset it
95          MOV R0, #30h
96          SJMP DONE
97
98  ; Leave the ISR
99  DONE:
100         NOP
101         RETI
102
103 ;---------------------EXTERNAL INTERRUPT 0 SERVICE ROUTINE
```

```
104 ; External interrupt service routine to detect pulses
105 ISREX0:
106         ; Check if timer 1 is running
107         JB TR1, CALCULATE_TIME
108
109         ; If not running, start the timer and reset the counter
110         SETB TR1
111         SJMP RESET_COUNTER
112
113         RETI
114
115 ; Calculate the time duration between pulses
116 ; (scaled down by factor of 5 to handle 8-bit arithmetic properly)
117 CALCULATE_TIME:
118         ; Stop timer 1 while calculations are done
119         CLR TR1
120         MOV A, R6           ; Move into acc number of 48ms intervals
121         SJMP GET_BPM
122
123 ; Routine to calculate the BPM according to the number of times the
        timer overflew between two pulses
124 GET_BPM:
125         ; Move the N of timer overflows into B to do division
126         MOV B, A                ; R6 is stored in the accumulator
127
128         ; Move 250 into A (250 = 1250/5 = (60,000/48)/5) so we get
              the division for BPM scaled down by a factor of 5. 48ms
              is the timer overflow delay
129         MOV A, #250             ; 250 we can only do division of 8
              bits (i.e. max numerator is 255)
130
131         ; Perform 8-bit division
132         DIV AB
133
134         ; Save quotient into R3
135         MOV R3, A
136
137         ;Save remainder into R4
138         MOV R4, B
139
140         ; SCALING BACK UP
141
142         ; Scale quotient
143         ; Move 5 into B to scale back up the quotient
144         MOV B, #05
145
```

```
146         ; Scale back up
147         MUL AB
148
149         ; Result should be within 0 and 255, so ignore higher byte
                of multiplication
150         MOV R1, A               ; Store the scaled quotient into R1
                for later addition.
151
152         ; Scale remainder
153         MOV A, R4               ; Move the remainder into acc
154         MOV B, #05              ; Move 5 into B to scale back up
155         MUL AB                  ; Multiply
156
157         MOV B, R6               ; Move the N of timer overflows into
                 B for proper scaling
158         DIV AB                  ; Divide the result of the
                multiplication by R6 to get remainder properly scaled
159
160         ; Add scaled remainder and scaled quotient to get BPM
161         ADD A, R1               ; A -> Scaled remainder. R1 ->
                Scaled quotient. A = A + R1
162         MOV BPM, A              ; Move the BPM into variable BPM
163
164
165         ; Decode the values of BPM to display properly
166         ACALL DECODE
167
168         ; Reset R6 to start the next measurement between pulses.
169         SJMP RESET_COUNTER
170
171
172 RESET_COUNTER:
173         MOV R6, #00             ; Reset pulse counter for next
                measurement
174         SETB TR1
175         RETI
176 ;
    ------------------------------------------------------------------------------


177
178 ; Special routine to move the result of BPM into something the
    display can read
179 DECODE:
180         ; Move the BPM into accumulator
181         MOV A, BPM
182         MOV B, #100             ; Move 100 into B to divide and get
                hundreds_bpm
```

```
183
184         ; Perform division. A -> result, B -> remainder
185         DIV AB
186         ADD A, #20h             ; Add 20H so it is in digit 2
187         MOV HUNDREDS_BPM, A     ; Move result into hundreds_bpm
188         MOV A, B                ; Move remainder into accumulator to
                now calculate tens_bpm
189         MOV B, #10              ; Move 10 into B to divide and get
               tens_bpm
190
191         ; Perform division to get tens_bpm
192         DIV AB
193         ADD A, #10h             ; Add 10H so it is digit 1
194         MOV TENS_BPM, A         ; Move result into tens_bpm
195         MOV A, B                ; Move remainder into accumulator to
                now calculate ones_bpm.
196         MOV B, #1               ; Move 1 into B to divide and get
               ones_bpm
197
198         ; This extra step in unnecessary, we could do MOV ONES_BPM,
               B so we get the remainder right away.
199         ; Perform division to get ones_bpm
200         DIV AB
201         MOV ONES_BPM, A
202         RET
203
204
205 ; -------------------TIMER 1 INTERRUPT SERVICE ROUTINE
      ----------------------------------------------------
206 ; Timer 1 ISR for calculating time between pulses
207 ISRTF1:
208         ; Clear interrupts and timer flag 1
209         CLR TR1
210         CLR TF1
211         CLR EA
212
213         ; Increment the times 10ms have occured after last pulse
214         INC R6
215
216         ; Reload timer
217         MOV TH1, #044h      ; 48 ms delay timer
218         MOV TL1, #07Fh      ; (FFFF - (48,000)d)
219
220         ; Restart interrupts, and restart timer
221         SETB EA
222         SETB TR1
223
```

```
224             RETI
225
226             END
```